

Multivariate Analysis of Variance (MANOVA) II: Practical Guide to ANOVA and MANOVA for SAS

Terminology for ANOVA

This chapter provides practical points in performing ANOVA and MANOVA. First, it is necessary to develop some terminology. Let us begin with the Kurlu example. The structure of the data would look like this:

Data Layout for the Kurlu Example							
Between Subjects		Within Subjects					
Group	Subject	Pretest			Posttest		
		SI	SF	OA	SI	SF	OA
1 (Control)	1						
.	.						
1 (Control)	10						
2 (Cognitive)	1						
.	.						
2 (Cognitive)	10						
3 (Behavioral)	1						
.	.						
3 (Behavioral)	10						
4 (Abreaction)	1						
.	.						
4 (Abreaction)	10						

The observations are the 40 patients who participated in the study. It is always recommended that the observations form the rows of the data matrix with one and only one row for an observation. For example, the data for the first observation in the Kurlu data set, Herkimer Schwatzbiggle, is given below.

Name	Group	Sub- ject	si_ pre	sf_ pre	oi_ pre	si_ post	sf_ post	oi_ post
Herkimer Schwatzbiggle	Control	1	46	68	70	72	74	82

In entering data into a database or spreadsheet, it is entirely legitimate to make two rows for Herk, one for his pre-test scores and the second for the post-test scores. The data would then be structured like this:

Name	Group	Subject	Time	si	sf	oa
Herkimer Schwatzbiggle	Control	1	Pre	46	68	70
Herkimer Schwatzbiggle	Control	1	Post	72	74	82

There is indeed nothing the matter with the structure in this table. One can perform ANOVAs and MANOVAs using this structure. However, setting up the ANOVA model for the structure in this table is much more difficult and error-prone than it is for the structure in the previous table. Consequently, for students learning these techniques, it is highly recommended to make certain that each row of the data matrix contains one and only one observational unit.

Returning to the data, we notice that the observations are organized into groups corresponding to the four therapies. The analysis will use the variable Group as the independent variable or predictor variable. In ANOVA terms, an independent variable that classifies individual observations into categories is called an ANOVA *factor*. The term “factor” in this sense should not be confused with a “factor” from factor analysis. In the Kurlu example, there is one and only one factor. When there is only one factor, the design is referred to as a *oneway* ANOVA.

When there is more than a single ANOVA factor, the design is called a *factorial* design. For example, suppose that the patients in the Kurlu study were subdivided into those who had previous treatment for the disorder and those who had no previous treatment. If the ANOVA model then used the presence or absence of prior treatment as an independent variable, the design would look like that in the following table.

	Therapy:			
Prior Treatment:	Control	Cognitive	Behavioral	Abreaction
No				
Yes				

Here, the ANOVA model would be referred to as a *twoway, factorial* design or sometimes just a *twoway* design. If there were three ANOVA factors, the design would be called a *threeway, factorial* ANOVA or simply a *threeway* ANOVA.

The individual groups within an ANOVA factor are referred to as the *levels* of the factor. For example, the Therapy factor in the above table has four levels (Control, Cognitive, Behavioral, and Abreaction) and the Previous Treatment factor as two levels (Yes and No). The term “level” does not necessarily imply that the groups are ordered according to some scale of magnitude. The term may simply refer to the different categories of an ANOVA factor without implication that one category has more of something than the next category. An ANOVA factor like sex, with its levels of female and male, would be an example of this.

A continuous variable entered into an ANOVA model is referred to as a *covariate*. Occasionally, when a covariate is used, the design is called *an analysis of covariance* or *ANCOVA*. In the Kurlu example, age of the patient might be used as a covariate.

In the Kurlu example, there were exactly 10 individuals in each of the four cells. When the number of individuals is identical in each cell of an ANOVA, the design is called a *balanced* or *orthogonal* design. When one or more cells have different numbers of individuals, then the design is called *unbalanced* or *nonorthogonal*. It is very important to determine whether a design is orthogonal or nonorthogonal before proceeding with the analysis. If a design is nonorthogonal, then there is more than one solution to the ANOVA table. It then becomes important to determine which of the solutions is preferable.

SAS Code for the Analysis of Variance

SAS has two basic procedures to use with ANOVA, and several more sophisticated procedures to deal with very specialized designs or problems with ANOVA. The two basic procedures are *PROC ANOVA* and *PROC GLM*, for General Linear Model. *PROC ANOVA* should be used only with balanced designs. *PROC GLM* may be used with either balanced or unbalanced designs. *PROC GLM* and *PROC ANOVA* both have the same syntax and will give identical results when the design is orthogonal. However, when the design is nonorthogonal, the *PROC ANOVA* usually will give incorrect results. Hence, to avoid errors it is recommended that one use *PROC GLM* and only *PROC GLM*.

The syntax for *PROC GLM* is

```
PROC GLM DATA =<data set name > ORDER = data
                                         formatted
                                         internal
                                         freq
```

Although the *ORDER=* option is not necessary, it is highly recommended that it be used in order to avoid errors. It specifies which level of the ANOVA factor is level number 1, which is level 2, etc. To see how this option operates, consider the following SAS program that will be used as input to an ANOVA where sex is the ANOVA factor..

```
PROC FORMAT;
    FORMAT sexfmt 1='Male' 2='Female';
DATA bagels;
    INPUT name sex bagels;
    LABEL bagels = 'Number of bagels consumed in five minutes';
    FORMAT sex sexfmt.;
CARDS:
    Waldo      1      6
    Esteretta  2      2
    Orestes    1     17
    Beulah     2      3
    Wilburina  2      1
```

;

If `ORDER = data` is used in the `PROC GLM` statement then level number 1 is the first level encountered in the data set. The first observation is Waldo, a male, so males will be the first level and females the second level for the ANOVA factor sex.

If `ORDER = formatted` is used then SAS uses the alphabetical order of the formatted variable used in the analysis. The variable sex is associated with the format `sexfmt`, so instead of printing a “1” SAS will print “Male” and instead of “2” SAS will print “Female.” Because F precedes M in the alphabet, females will be the first level and males will be the second level of the ANOVA factor sex.

If `ORDER = internal` then SAS does not use the formatted labels. Instead, it orders by the numeric values of the variable. The ordering for sex in this case is 1 and then 2, so males will be the first level and females the second level.

If `ORDER = freq` then SAS will order the levels in terms of decreasing frequency in the data. Because there are three females and two males, the first level of sex will be female and the second will be male.

The following statements appear after the `PROC GLM` statement:

```
CLASS <names of the variables to be used as ANOVA factors in the model> ;
MODEL <dependent variable(s)> = <independent variable(s)> ;
```

In the bagel data set, for example, the commands could read

```
PROC GLM DATA=bagels ORDER=formatted;
      CLASS sex;
      MODEL bagels = sex;
RUN;
```

In the Kurlu data set, the following commands could be used

```
PROC GLM DATA=kurlu ORDER=data;
      CLASS group;
      MODEL si_pre sf_pre oi_pre = group;
RUN;
```

This statement would perform three different ANOVA, one for each of the three variables in the dependent variable list.

In factorial designs, interactions are designated by placing a star (*) between the variables in the interaction. If an interaction term is not explicitly states, then SAS will ignore that interaction. For example, if the previous treatment variable for the Kurlu problem was called `pretreat`, then the following statement will only fit the main effects for therapy group and for prior treatment:

```
CLASS group pretreat;
MODEL si_pre si_post = group pretreat;
```

The following statements would also fit the interaction term:

```
CLASS group pretreat;
MODEL si_pre si_post = group pretreat group*pretreat;
```

A vertical bar (|) is a shortcut for specifying “all the interactions” between the variables that surround the bar. For example,

```
CLASS sex age religion;
MODEL attitude = sex | age | religion;
```

would fit the three main effects for sex, age, and religion, the three different two-way interactions (sex*age, sex*religion, and age*religion), and the three way interaction of sex*age*religion. On the other hand, the following statements

```
CLASS sex age religion;
MODEL attitude = sex | age religion;
```

would fit the three main effects but only the interaction between sex and age.

To place a covariate into the analysis, simply enter it into the list of independent variables on the MODEL statement. For example,

```
CLASS group;
MODEL si_pre = group age;
```

will treat age as a covariate in the Kurlu example.

The MEANS statement for PROC GLM prints out the group means and standard deviations for one or more ANOVA effects. The syntax is

```
MEANS <list of ANOVA effects>;
```

For example the statement

```
MEANS sex religion sex*religion;
```

would print out three tables. The first would give the means for sex, the second would give the means for religion, and the third would give the means for all combinations of sex and religion.

Sums of Squares in ANOVA

In an orthogonal or balanced ANOVA in which there are equal numbers of observations in each cell of the ANOVA design, there is no need to worry about the decomposition of sums of squares. Here, one ANOVA factor is completely uncorrelated with another ANOVA factor, so a test for, say, a sex effect is independent of a test for,

say, an age effect. Completely balanced designs like this are usually obtainable in preplanned experiments.

When the design is unbalanced or nonorthogonal (i.e., the number of observations vary from one cell to another), then there is not a unique decomposition of the sums of squares. Here, the effects for one ANOVA factor may be correlated with those for another ANOVA factor. Hence, decisions must be made to account for the correlation between the ANOVA factors in terms of quantifying the effects of any single factor. The situation is mathematically equivalent to a multiple regression model where there are correlations among the predictor variables. Each variable has direct and indirect effects on the dependent variable. In an ANOVA, each ANOVA factor will have direct and indirect effects on the dependent variable.

SAS can print out four different types of sums of squares. In an orthogonal design, all four will be equal. In a nonorthogonal design, the “correct” sums of squares will depend upon the logic of the design. To illustrate these four sums of squares, consider the following statement:

```
PROC GLM DATA=attitudes ORDER=internal;
    CLASS sex religion;
    MODEL attitude = sex | religion;
RUN;
```

The first sums of squares are called **Type I** sums of squares by SAS. This performs an hierarchical decomposition. That is, the sum of squares for sex is calculated first, then the sum of squares for religion, controlling for sex, is calculated next, and finally, the sum of squares for the interaction of sex and religion is calculated, controlling for the main effect of sex and then the main effect of religion given sex.

The second sums of squares are called **Type II** sums of squares and is equal to the sums of squares from a multiple regression of the dependent variable on the quantified equivalent of the variables sex, religion, and the sex *religion interaction. The sum of squares for sex controls for religion and the sex*religion interaction, the sum of squares for religion controls for sex and the sex*religion interaction, and the sum of squares for sex*religion controls for sex and for religion.

The third sums squares are called **Type III** sums of squares. The decomposition of the sum of squares for an effect is identical to that of Type II sums of squares. For example, the sum of squares for sex is adjusted for the effects of religion and for the interaction of sex and religion. However, Type III sums of squares adjusts the sums of squares to guesstimate what they might be if the design were truly orthogonal. To illustrate the difference between Type II and Type III SS, consider the factor sex. If the data had 60% females and 40% males, then Type II sums of squares simply makes its estimates based on a sample of 60% females and 40% males. Type III SS assumes that the sex difference came about because of sampling and tries to generalize to a population in which the number of males and females is equal.

The fourth sums of squares, **Type IV**, is identical to Type III SS, but should be used whenever there is a missing cell in the ANOVA. For example, if by dumb luck there

are no male Episcopalians in the sample, then Type IV SS adjusts the sums of squares for that fact.

The SAS default provides both Type I and Type III sums of squares. If you wish any others, specify the SS2 or SS4 options on the model statement. For example, the statement

```
MODEL attitude = sex | religion / SS1 SS2 SS3 SS4;
```

will provide all four types of sums of squares.

A priori and a posteriori tests of means.

The F statistic from an analysis of variance simply tells whether the means for an ANOVA effect are within random sampling error of one another. The null hypothesis states that there is no effect, so the means for the effect should be random samplings of the same distribution of means (or the same “hat” of means). The F statistic and its associated p level give a quantitative index of how likely it is that the means are really being pulled out of the same sampling distribution or the same “hat.” If the F value is large and the associated p value is small, then it is unlikely that the means are within sampling error of one another. In other words, it is likely that the means are being pulled from different “hats” and that there are true differences somewhere among the means.

What the F statistic does not say is *where* the real differences among the means exists. To illustrate this, return to the Kurlu examples. There are four groups. The F statistic for a variable such as social functioning, tells whether the four means on this variable are within sampling error of one another. The statistic does not tell us which means differ from which other means. Consider the following SAS statements:

```
LIBNAME p7291dir '~carey/p7291dir';
PROC GLM DATA=p7291dir.kurlu ORDER=internal;
  CLASS therapy;
  MODEL sf_post = therapy;
  MEANS THERAPY;
RUN;
```

The output from this statement is given in the table below.

```
KURLU Example
General Linear Models Procedure
Class Level Information

Class      Levels      Values
THERAPY      4      Abreaction Behavioral Cognitive Control

Number of observations in data set = 40

--- <PAGE> -----
KURLU EXAMPLE
General Linear Models Procedure
```

Dependent Variable: SF_POST Post Social

Source	DF	Sum of Squares	F Value	Pr > F
Model	3	1522.27500000	4.51	0.0087
Error	36	4052.50000000		
Corrected Total	39	5574.77500000		

R-Square	C.V.	SF_POST Mean
0.273065	17.02347	62.3250000

Source	DF	Type I SS	F Value	Pr > F
THERAPY	3	1522.27500000	4.51	0.0087

Source	DF	Type III SS	F Value	Pr > F
THERAPY	3	1522.27500000	4.51	0.0087

--- <PAGE> -----
 KURLU Example
 General Linear Models Procedure

Level of THERAPY	N	Mean	SD
Abreaction	10	68.0000000	11.1455023
Behavioral	10	66.2000000	10.4965603
Cognitive	10	63.0000000	8.3931189
Control	10	52.1000000	12.0595743

It is clear from the ANOVA that there are true differences among the means. The F value is 4.51 and its associated p level is less than .01. Visual examination of the means suggests that the lowest social functioning group is the control group, and that all three experimental therapies have higher average levels of social functioning than the controls. From this observation, one might conclude that the experimental therapies appear to differ significantly from the controls. However, this conclusion--while it may indeed be true--is not justified from the data. For example, it could be that the real difference is between the Abreaction group and the Control group. How does one test for this?

Statisticians eschew the approach that most students of statistics might first think of to solve this problem--performing individual t -tests among all pairs of means. The reason is that there would be six different t -tests, so the chance of a Type I error (rejecting the null hypothesis of no mean difference when in fact, there is no mean difference) is increased. There are two approaches that statisticians use. They are the *a priori* approach and the *a posteriori* (also known as *post hoc*) approach.

A posteriori or post hoc tests

There are many different *a posteriori* or *post hoc* tests, and it is not the province of this chapter to discuss them all. What these tests have in common is an attempt to arrive at a minimum value for the difference between two means that would make the

means differ significantly according to some adjusted α levels. In SAS, all of the *post hoc* tests are performed using the MEANS statement. For example, the statement

```
MEANS therapy / DUNCAN;
```

will perform Duncan's multiple range test. The output from this procedure is given below.

KURLU Example

General Linear Models Procedure

Duncan's Multiple Range Test for variable: SF_POST

NOTE: This test controls the type I comparisonwise error rate, not the experimentwise error rate

Alpha= 0.05 df= 36 MSE= 112.5694

Number of Means	2	3	4
Critical Range	9.62	10.12	10.44

Means with the same letter are not significantly different.

Duncan Grouping	Mean	N	THERAPY
A	68.000	10	Abreaction
A	66.200	10	Behavioral
A	63.000	10	Cognitive
B	52.100	10	Control

Here, the rows labeled Number of Means and Critical Range give the difference between means that would be significant for a comparison of k group means. For example, two groups would have to have means differing by 9.62 units to be significantly different. For three groups, the mean differences among all three groups would have to be at least 10.12 units for each of the three group means to be different. That is, group 1's mean is significantly different from group 2's mean which, in turn, is significantly different from group 3's mean when the groups are ordered according to the mean. The results of Duncan's test suggest that the three experimental groups do not differ among one another. The Control group, however, differs significantly from all three experimental groups.

Another popular post hoc test is the Scheffe test. The SAS statement

```
MEANS therapy / SCHEFFE LINES;
```

produces the following output

KURLU Example

General Linear Models Procedure

Scheffe's test for variable: SF_POST

NOTE: This test controls the type I experimentwise error rate but generally has a higher type II error rate than REGWF for all pairwise comparisons

Alpha= 0.05 df= 36 MSE= 112.5694
 Critical Value of F= 2.86627
 Minimum Significant Difference= 13.914

Means with the same letter are not significantly different.

Scheffe Grouping	Mean	N	THERAPY
A	68.000	10	Abreaction
A	66.200	10	Behavioral
B A	63.000	10	Cognitive
B	52.100	10	Control

For this test, two means must differ by 13.914 units to be considered significantly different. Again, the three experimental therapies do not differ among one another. However, in this test, the Cognitive therapy does not differ significantly from the Control group. Thus, the Scheffe test gives different substantive results from Duncan's test. This situation is not uncommon when a number of different *post hoc* tests are applied to the same data.

***A priori* tests: contrast coding**

A priori tests are made before the fact. That is, before the ANOVA is performed, the researcher has one or more hypotheses about the group means and then deliberately codes the data to test this hypothesis. For example, a natural hypothesis in the Kurlu example is whether the three experimental therapies on average do better than the control group.

Contrast coding permits this by literally creating a new independent variable from the levels of an ANOVA factor. The general form of a contrast code is

$$\begin{array}{ccc} \text{New} & & \text{Levels of} \\ \text{ANOVA} & & \text{ANOVA} \\ \text{Independent} & = & \text{Independent} \\ \text{Variable(s)} & & \text{Variable} \end{array} \quad \begin{array}{c} \text{Matrix} \\ \text{of} \\ \text{Contrast Codes} \end{array}$$

The only requirement for the contrast codes is that the coefficients in each row sum to 0. For example, to test the null hypothesis that the average of the three experimental therapies does not differ from the control therapy, the contrast code would be

$$\begin{array}{r} \text{Abreaction} \\ \text{Behavioral} \\ \text{Cognitive} \\ \text{Control} \end{array} \left(\text{Experimental vs Control} \right) = \begin{pmatrix} -1 & -1 & -1 & 3 \end{pmatrix}$$

This statement is mathematically equivalent to testing the null hypothesis

$$0 = -1\mu(\text{Abreaction}) - 1\mu(\text{Behavioral}) - 1\mu(\text{Cognitive}) + 3\mu(\text{Control}).$$

In SAS, the CONTRAST statement is used for contrast coding. The syntax for the CONTRAST statement is

```
CONTRAST '<label for printing>' <name of ANOVA effect> <contrast codes>;
```

For example, to test whether the control group differs significantly from the mean of the three experimental therapies, the statement would be

```
CONTRAST 'Control vs. Xpermntl' therapy -1 -1 -1 3;
```

The contrast codes in this example sum to zero so they are legitimate values. The codes in the above statement are given in terms of the ORDER= option on the PROC GLM statement. **Consequently, it is exceptionally important to make certain that the numbers in the CONTRAST statement agree with the ordering of the levels of an ANOVA factor. For this reason, it is always good practice to specify the ORDER= option in the PROC GLM statement.** Because the order of the groups in this example is alphabetical, the three experimental therapies are first and the control therapy is last.

The following statements illustrate the use of this with SAS using the Social Functioning Posttest score as the dependent variable

```
LIBNAME p7291dir '~carey/p7291dir';
PROC GLM DATA=p7291dir.kurlu ORDER=internal;
  CLASS therapy;
  MODEL sf_post = therapy;
  CONTRAST 'Control vs. Xpermntl' therapy -1 -1 -1 3;
RUN;
```

Note that placing the CONTRAST statement before the RUN statement makes this an a priori contrast. If one first looked at the ANOVA results and the means and then developed a contrast hypothesis on the basis of that, then the hypothesis test is no longer a priori. Adding this contrast statement to the PROC GLM procedure produces the following output.

```
KURLU EXAMPLE
General Linear Models Procedure
```

Dependent Variable: SF_POST Post Social

Source	DF	Sum of Squares	F Value	Pr > F
Model	3	1522.27500000	4.51	0.0087
Error	36	4052.50000000		
Corrected Total	39	5574.77500000		

R-Square	C.V.	SF_POST Mean
0.273065	17.02347	62.3250000

Source	DF	Type I SS	F Value	Pr > F
THERAPY	3	1522.27500000	4.51	0.0087

Source	DF	Type III SS	F Value	Pr > F
THERAPY	3	1522.27500000	4.51	0.0087

Contrast	DF	Contrast SS	F Value	Pr > F
Control vs Xpermnt1	1	1394.00833333	12.38	0.0012

The chief advantage of the contrast hypothesis test as opposed to the simple F test from the ANOVA is that it is a more powerful test of the hypothesis that the experimental therapies work. In the contrast hypothesis, there is a single degree of freedom while in the ANOVA there are three degrees of freedom for the F ratio. This increase in power is apparent from the lower p level for the contrast.

To illustrate how the CONTRAST statement is equivalent to creating a new variable, try running the following SAS program and compare its output to the one given above.

```
LIBNAME p7291dir '~carey/p7291dir';
DATA temp;
  SET p7291dir.kurlu;
  IF therapy='Control' THEN convsxpr = 3;
  ELSE convsxpr=-1;
  LABEL convsxpr='Control vs Xpermnt1';
RUN;
PROC GLM DATA=temp;
  MODEL sf_post = convsxpr;
RUN;
```

More detail about contrast coding is provided in the Appendix which will become available whenever I get enough time to write the damn thing.

Polynomial Contrast Codes

One strongly recommended use of contrast codes is for ANOVA factors where the levels of the factor are ordered according to some scale of magnitude. To illustrate this, consider a new data set (on ~carey/p7291dir/political.attitudes.sas) on political attitudes. The dependent variables in the study are all measures of liberalism versus conservatism where high scores are associated with more liberal attitudes. The first dependent variable measures attitudes towards abortion, the second dependent variable

measures attitudes toward affirmative action, and the third dependent variable measures attitudes toward health care reform.

There are two independent variables in the study. The first is gender of the respondent (female or male), and the second is education. Education is divided into three levels--some high school, high school graduate, and college graduate--and individuals are placed into the highest category achieved. The design is shown below.

Design of the Political Attitudes Study			
	Education:		
Sex:	< high school grad	high school grad	college grad
Male			
Female			

This is a straight-forward, twoway ANOVA with sex as one factor and education as the other factor. Sex has two levels and education has three levels, so the design could be called a 2 by 3 ANOVA. The usual SAS statements to analyze the data would be

```
PROC GLM ORDER=internal;
  CLASS sex educ;
  MODEL att1 att2 att3 = sex | educ;
RUN;
```

However, education is an ordered variable. Group 2 (high school grads) has more formal education than group 1 and group 3 (college grads) has more formal education than group 2. If attitudes are associated with education, one can arrive at a more powerful test by contrast coding education into a linear effect and a quadratic effect. This is called a polynomial contrast code. The GLM statements would read:

```
PROC GLM ORDER=internal;
  CLASS sex educ;
  MODEL att1 att2 att3 = sex | educ;
  CONTRAST 'educ: linear' educ -1 0 1;
  CONTRAST 'educ: quadratic' educ -1 2 -1;
RUN;
```

In the ordinary ANOVA, there would be two degrees of freedom associated with education. The contrast, however, splits these into two tests, each with a single degree of freedom. The first test, the linear contrast, tests whether attitudes change linearly with education. Because education has three levels, this test is equivalent to testing whether the mean of college grads differs from that of high school dropouts. The second test, the quadratic, literally tests whether the mean for the middle group of high school grads differs significantly from the average of the means for high school dropouts and college grads. The results of these tests for two variables, att2 and att3, are given below.

Political Attitudes: sex & education
General Linear Models Procedure

Dependent Variable: ATT2 Affirmative Action

Source	DF	Sum of Squares	F Value	Pr > F
Model	5	1085.14166667	3.25	0.0088
Error	114	7613.65000000		
Corrected Total	119	8698.79166667		


R-Square	C.V.	ATT2 Mean
0.124746	20.97701	38.9583333

Source	DF	Type I SS	F Value	Pr > F
SEX	1	7.00833333	0.10	0.7466
EDUC	2	992.11666667	7.43	0.0009
SEX*EDUC	2	86.01666667	0.64	0.5271

Source	DF	Type III SS	F Value	Pr > F
SEX	1	7.00833333	0.10	0.7466
EDUC	2	992.11666667	7.43	0.0009
SEX*EDUC	2	86.01666667	0.64	0.5271

Contrast	DF	Contrast SS	F Value	Pr > F
Educ: linear	1	644.11250000	9.64	0.0024
Educ: quadratic	1	348.00416667	5.21	0.0243

The significance of the linear contrast suggests that the means for college grads differ from those of high school dropouts on attitudes towards affirmative action. The significance of the quadratic contrast suggests that the means for high school grads does not lie midway between the means for the high school dropouts and the college grads. As in any ANOVA, the means must be inspected to tell us in which direction these differences fall. The graph below depicts the means for males and females on attitudes toward affirmative action as a function of education.



Conclusion here would be that education is associated with more liberal attitudes toward affirmative action. But this is apparent only for college graduates. High school graduates have the same level of attitudes as high school dropouts.

The results for the variable att3 (attitudes toward health care) are given below.

Political Attitudes: sex & education
General Linear Models Procedure

Dependent Variable: ATT3 Health Care

Source	DF	Sum of Squares	F Value	Pr > F
Model	5	222.27500000	1.34	0.2518
Error	114	3777.05000000		
Corrected Total	119	3999.32500000		


R-Square 0.055578 C.V. 20.24993 ATT3 Mean 28.4250000

Source	DF	Type I SS	F Value	Pr > F
SEX	1	37.40833333	1.13	0.2902
EDUC	2	171.80000000	2.59	0.0792
SEX*EDUC	2	13.06666667	0.20	0.8213

Source	DF	Type III SS	F Value	Pr > F
SEX	1	37.40833333	1.13	0.2902
EDUC	2	171.80000000	2.59	0.0792
SEX*EDUC	2	13.06666667	0.20	0.8213

Contrast	DF	Contrast SS	F Value	Pr > F
Educ: linear	1	156.80000000	4.73	0.0317
Educ: quadratic	1	15.00000000	0.45	0.5024

Here, the ANOVA results for education suggest that no significant differences, although the p level does suggest a trend. The linear contrast, however, is significant while the quadratic contrast is not. This suggests that there is indeed an association between education and health care attitudes, but using a two degrees of freedom test (ANOVA) as opposed to a single degree of freedom test (linear contrast) hides the relationship. Again, the means, given in the following graph, must be inspected to detail in which direction these differences lie.



MANOVA: Multivariate Analysis of Variance

Understanding MANOVA requires understanding of sampling, a process outlined in the handout on sampling and one that deserves some repetition here. Although we often speak of sampling “scores” or “numbers” and refer to these scores as being pulled randomly from a hat, in actuality, we sample *observations*, not scores. When running an experiment one literally has an object (a person, a rat, a tree, etc.) that has a whole list of attributes. In univariate ANOVA, we are interested in only one attribute, so we can think in terms of the quantification of that attribute into a “score.”

In reality, what we are doing is taking an observation and ignoring all those attributes of the observation except for the single attribute of interest. That single attribute is the dependent variable. MANOVA depends upon the understanding that we sample an observation, and then ignore all those attributes except for the *two or more* attributes of interest. Those two or more attributes are the dependent variables. Hence, instead of loosely talking about a single score, MANOVA loosely talks about a *vector* of scores.

To complete the analogy, the ANOVA for a single variable in the Kurlu example tests whether the scores on that variable for the four groups can be regarded as being

pulled out of the same “hat” of scores. A MANOVA for the Kurlu example tells us whether the *vectors* of scores for the four variables may be regarded as being pulled out of the same “hat” of *vectors*. Return to the Kurlu example and examine write the GLM procedure for all three variables after treatment:

```
LIBNAME here '';
OPTIONS NOCENTER NONUMBER NODATE LINESIZE=64;
TITLE KURLU Example;
PROC GLM DATA=here.kurlu ORDER=internal;
  CLASS therapy;
  MODEL si_post sf_post oi_post = therapy;
  CONTRAST 'Contrl vs Xpermntl' therapy -1 -1 -1 3;
  MANOVA H=therapy / PRINTE;
RUN;
```

The output from these statements is given below.

KURLU Example

General Linear Models Procedure

Class Level Information

Class	Levels	Values
THERAPY	4	Abreaction Behavioral Cognitive Control

Number of observations in data set = 40

--- <PAGE> -----

KURLU Example

General Linear Models Procedure

Dependent Variable: SI_POST		Post Symptoms		
Source	DF	Sum of Squares	F Value	Pr > F
Model	3	297.27500000	1.07	0.3743
Error	36	3336.50000000		
Corrected Total	39	3633.77500000		

R-Square	C.V.	SI_POST Mean
0.081809	16.43547	58.5750000

Source	DF	Type I SS	F Value	Pr > F
THERAPY	3	297.27500000	1.07	0.3743

Source	DF	Type III SS	F Value	Pr > F
THERAPY	3	297.27500000	1.07	0.3743

Contrast	DF	Contrast SS	F Value	Pr > F
Contrl vs Xpermntl	1	279.07500000	3.01	0.0912

--- <PAGE> -----

KURLU Example

General Linear Models Procedure

Dependent Variable: SF_POST

Post Social

Source	DF	Sum of Squares	F Value	Pr > F
Model	3	1522.27500000	4.51	0.0087
Error	36	4052.50000000		

Corrected Total	39	5574.77500000		
	R-Square	C.V.	SF_POST Mean	
	0.273065	17.02347	62.3250000	
Source	DF	Type I SS	F Value	Pr > F
THERAPY	3	1522.27500000	4.51	0.0087
Source	DF	Type III SS	F Value	Pr > F
THERAPY	3	1522.27500000	4.51	0.0087
Contrast	DF	Contrast SS	F Value	Pr > F
Contrl vs Xpermntl	1	1394.00833333	12.38	0.0012

--- <PAGE> -----

KURLU Example
General Linear Models Procedure

Dependent Variable: OI_POST Post Occup

Source	DF	Sum of Squares	F Value	Pr > F
Model	3	225.07500000	0.65	0.5896
Error	36	4170.70000000		
Corrected Total	39	4395.77500000		

	R-Square	C.V.	OI_POST Mean
	0.051203	19.28078	55.8250000

Source	DF	Type I SS	F Value	Pr > F
THERAPY	3	225.07500000	0.65	0.5896

Source	DF	Type III SS	F Value	Pr > F
THERAPY	3	225.07500000	0.65	0.5896

Contrast	DF	Contrast SS	F Value	Pr > F
Contrl vs Xpermntl	1	99.00833333	0.85	0.3614

E = Error SS&CP Matrix

	SI_POST	SF_POST	OI_POST
SI_POST	3336.5	1607.8	1626.8
SF_POST	1607.8	4052.5	2309.9
OI_POST	1626.8	2309.9	4170.7

--- <PAGE> -----

KURLU Example
General Linear Models Procedure
Multivariate Analysis of Variance
Partial Correlation Coefficients from the Error SS&CP Matrix / Prob > |r|

DF = 36	SI_POST	SF_POST	OI_POST
SI_POST	1.000000	0.437245	0.436098
	0.0001	0.0068	0.0070
SF_POST	0.437245	1.000000	0.561859
	0.0068	0.0001	0.0003
OI_POST	0.436098	0.561859	1.000000
	0.0070	0.0003	0.0001

--- <PAGE> -----

KURLU Example

General Linear Models Procedure
Multivariate Analysis of Variance

Characteristic Roots and Vectors of: E Inverse * H, where
H = Type III SS&CP Matrix for THERAPY E = Error SS&CP Matrix

Characteristic Root	Percent	Characteristic Vector V'EV=1	
		SI_POST	SF_POST
0.41287135	89.11	0.00206074	0.01755008
		0I_POST	
		-0.00605616	
0.04845571	10.46	-0.01551421	0.00200294
		0.01451824	
0.00199145	0.43	0.01231271	-0.00862198
		0.01129095	

Manova Test Criteria and F Approximations for
the Hypothesis of no Overall THERAPY Effect
H = Type III SS&CP Matrix for THERAPY E = Error SS&CP Matrix
S=3 M=-0.5 N=16

Statistic	Value	F	Num DF	Den DF	Pr > F
Wilks' Lambda	0.673726	1.6228	9	82.898	0.1221
Pillai's Trace	0.340425	1.5360	9	108	0.1444
Hotelling-Lawley Trace	0.463319	1.6817	9	98	0.1036
Roy's Greatest Root	0.412871	4.9545	3	36	0.0056

NOTE: F Statistic for Roy's Greatest Root is an upper bound.

Characteristic Roots and Vectors of: E Inverse * H, where
H = Contrast SS&CP Matrix for Contrl vs Xpermntl
E = Error SS&CP Matrix

Characteristic Root	Percent	Characteristic Vector V'EV=1	
		SI_POST	SF_POST
0.39778216	100.00	0.00296313	0.01744071
		0I_POST	
		-0.00703242	
0.00000000	0.00	-0.00431561	-0.00287712
		0.01804125	
0.00000000	0.00	0.01921291	-0.00859648
		0.00000000	

--- <PAGE> -----

KURLU Example
General Linear Models Procedure
Multivariate Analysis of Variance

Manova Test Criteria and Exact F Statistics for
the Hypothesis of no Overall Contrl vs Xpermntl Effect
H = Contrast SS&CP Matrix for Contrl vs Xpermntl
E = Error SS&CP Matrix
S=1 M=0.5 N=16

Statistic	Value	F	Num DF	Den DF	Pr > F
Wilks' Lambda	0.715419	4.5082	3	34	0.0091

Pillai's Trace	0.284581	4.5082	3	34	0.0091
Hotelling-Lawley Trace	0.397782	4.5082	3	34	0.0091
Roy's Greatest Root	0.397782	4.5082	3	34	0.0091

--- <PAGE> -----

The SAS output gives the univariate ANOVA for each of the dependent variables specified in the MODEL statement. Because a CONTRAST statement was given, SAS will also test the contrast of the control therapy versus the mean of the three experimental therapies for each of the dependent variables. Notice how the p level for the contrast statement is lower than that for the ANOVA for each of the three therapies. Examination of each ANOVA gives an equivocal test of the therapies. There is no evidence for therapeutic differences for two of the variables, Symptom Index and Occupational Adjustment while there is favorable evidence for some differences in Social Functioning. Could it be that the result for Social Functioning is a false positive? Or perhaps, one or both of the results for the Symptom Index and Occupational Adjustment are false negatives? A MANOVA can help to answer that question.

The MANOVA statement performs the multivariate analysis of variance. The H= subcommand specifies the ANOVA factor to test. Because this is a one-way ANOVA, there is only one ANOVA factor, therapy. The option PRINTE requests that the procedure print out the sums of squares and cross products (SS&CP) matrix for error and its associated correlation matrix.

The first output from MANOVA is the SS&CP matrix for the error term. Because this matrix is unscaled, there is no need to visually inspect it. The correlation matrix is more informative. It is called a "partial" correlation matrix because it controls for mean differences among the therapies. As in multivariate regression, if the independent variables predicted so much of the dependent variables that the remainder is, in fact, random error, then all correlation should be close to 0.0. The fact that these correlations deviate from 0 inform us that, within a group, individuals with high scores on the symptom index also tend to have high scores on social functioning, etc.

The next section of the output gives the eigenvalues (termed characteristic roots in the output) and eigenvectors (characteristic vectors) of the product of the inverse of the error SSCP matrix and the hypothesis SSCP matrix, or $\mathbf{E}^{-1}\mathbf{H}$. All the hypotheses tests for MANOVA are made on this matrix, so apparently some SAS programmer felt an overwhelming compulsion to print it out. For most people this part of the output is as interesting as a random collection of social security numbers.

The following section gives the results of the hypothesis test. There are four different test statistics, each with its own associated F statistic. In some designs, these four will give identical results. But in most cases--the present example being one--they will differ. Of the four, Pillai's trace is the most robust (i.e., least sensitive to departures from the assumptions). Wilk's Lambda (), however, is more often reported because the quantity $1 - \lambda$ gives the proportion of generalized variance in the dependent variables explained by the model. The two other test statistics--Hotelling-Lawley's trace and Roy's Greatest Root--are seldom used. Usually, Pillai's trace, Wilks' , and Hotelling-

Lawley trace give similar results. Roy's root is an upper bound limit to the F statistic, so it may give a very different F and p -value than the other three statistics. When this occurs, it is prudent to ignore Roy's statistic.

All of the statistics try to answer the following question: How likely is it that the 3 by 1 column vector of means for the four groups are being sampled from the same "hat" of 3 by 1 column vectors? To rephrase the question, are there differences in the 3 by 1 column vectors of means somewhere among these four groups? The F statistics and their associated p -values suggest a weak trend in this direction.

When a CONTRAST statement is given before the MANOVA statement, the MANOVA automatically tests the contrast. In this case, the contrast would test whether the 3 by 1 vector of means for the Control group differs significantly from the 3 by 1 vector of means averaged over the experimental therapies. All four test statistics give the same answer to this hypothesis. Because the F s are significant, it is highly likely that the means for the experimental treatments differ, on average, from the means for the Control group. Once again, we can see the advantage of using contrast coding for testing hypotheses.

Transformations

Considerable time has been spent discussing contrast coding in ANOVA because the major utility of MANOVA lies in an analogous coding scheme. Contrast coding is used to test hypotheses about *independent variables* or the variables on the right hand side of the ANOVA model. A major use of MANOVA is to do analogous contrasting to *dependent variables* or the variables on the left hand side of the ANOVA model. To avoid equivocation in the use of the word contrast, the term *transformation* will be used to denote coding of the *dependent* variables to test hypotheses about the variables.

A contrast of an independent variable literally creates a new independent variable and then analyzes the dependent variable(s) using this new independent variable. A transformation of dependent variables literally creates a new dependent variable and then analyzes the new dependent variable. The general form of a transformation is

$$\begin{array}{ccccc} \text{Vector} & & & & \text{Vector} \\ \text{of} & & & & \text{of} \\ \text{New} & = & \text{Transformation} & & \text{Old} \\ \text{Dependent} & & \text{Matrix} & & \text{Dependent} \\ \text{Variables} & & & & \text{Variables} \end{array}$$

To illustrate, return to the Kurlu example. Thus far, we have only analyzed the three variables measured after therapy. This was done for didactic reasons. A preferable type of analysis would be to control for baseline scores and then test for different outcomes on the post-test scores. For the Symptom Index, one possible approach is to create a new variable that subtracts the baseline from the post-test score. The following code illustrates this:

```

TITLE KURLU Example;
DATA temp;
  SET here.kurlu;
  si_diff = si_post - si_pre;
RUN;
PROC GLM DATA=temp ORDER=internal;
  CLASS therapy;
  MODEL si_diff = therapy;
  CONTRAST 'Control vs Exprmntl' therapy -1 -1 -1 3;
RUN;

```

Here, the analysis is for the new variable `si_diff`. The output for this example is:

```

----<PAGE>-----
KURLU Example
General Linear Models Procedure
Class Level Information

Class      Levels      Values
THERAPY    4      Abreaction Behavioral Cognitive Control

Number of observations in data set = 40
----<PAGE>-----
KURLU Example
General Linear Models Procedure

Dependent Variable: SI_DIFF

Source              DF      Sum of Squares      F Value      Pr > F
Model                3          481.4000000          2.27      0.0970
Error                36         2545.0000000
Corrected Total      39         3026.4000000

              R-Square              C.V.              SI_DIFF Mean
              0.159067              102.5366              8.2000000

Source              DF      Type I SS      F Value      Pr > F
THERAPY              3          481.4000000          2.27      0.0970

Source              DF      Type III SS      F Value      Pr > F
THERAPY              3          481.4000000          2.27      0.0970

Contrast              DF      Contrast SS      F Value      Pr > F
Control vs Exprmntl  1          388.8000000          5.50      0.0247
----<PAGE>-----

```

Compare this output with that for the variable `si_post` given in previous output. Notice how the p -value becomes smaller. The reason for this is that difference scores have a smaller variance than the original scores when the two original scores are positively correlated. This reduces the error variance which is the denominator in the F ratio. A smaller denominator increases the value of F . Hence, there is almost always increased power to detect effects by controlling for baseline measurements. Once again, we see the

advantage of using contrast codes to test for the efficacy of the three experimental therapies.

The C ONTRAST statement in PROC GLM saves the trouble of creating new independent variables in a DATA step. Likewise the M= option on the MANOVA statement saves the trouble of creating new dependent variables. The M= option, however, has a different syntax. Here one does not have to enter numbers for transformation codes (although one can do so, if desired). Instead one can simply express the algebraic equivalent using the original dependent variable names. For example, the following MANOVA performs the same analysis as the one that created a new variable, `si_diff`, above:

```
PROC GLM DATA=here.kurlu ORDER=internal;
  CLASS therapy;
  MODEL si_pre si_post = therapy;
  CONTRAST 'Control vs Xpermntl' therapy -1 -1 -1 3;
  TITLE2 MANOVA for difference scores;
  MANOVA H=therapy M=si_post - si_pre;
RUN;
```

Here, the M= option on the MANOVA statement gives the transformation of the dependent variables. In this case, the MANOVA will be performed on a “new variable” that equals the difference between post-test and pretest scores on the Symptom Index. The relevant output from this transformation is given below.

```
----<PAGE>-----
KURLU Example
MANOVA for difference scores

General Linear Models Procedure
Multivariate Analysis of Variance

M Matrix Describing Transformed Variables

          SI_PRE          SI_POST
MVAR1          -1          1
----<PAGE>-----
KURLU Example
MANOVA for difference scores

General Linear Models Procedure
Multivariate Analysis of Variance

Characteristic Roots and Vectors of: E Inverse * H, where
H = Type III SS&CP Matrix for THERAPY   E = Error SS&CP Matrix

Variables have been transformed by the M Matrix

Characteristic   Percent   Characteristic Vector   V'EV=1
      Root
          0.18915521   100.00   0.01982239
                                MVAR1
```

Manova Test Criteria and Exact F Statistics for
the Hypothesis of no Overall THERAPY Effect

on the variables defined by the M Matrix Transformation
 H = Type III SS&CP Matrix for THERAPY E = Error SS&CP Matrix

```
S=1      M=0.5      N=17
Statistic          Value          F      Num DF   Den DF   Pr > F
Wilks' Lambda      0.840933    2.2699    3        36    0.0970
Pillai's Trace     0.159067    2.2699    3        36    0.0970
Hotelling-Lawley Trace 0.189155    2.2699    3        36    0.0970
Roy's Greatest Root 0.189155    2.2699    3        36    0.0970
```

---<Section of output deleted> -----

Manova Test Criteria and Exact F Statistics for
 the Hypothesis of no Overall Control vs Xpermntl Effect
 on the variables defined by the M Matrix Transformation
 H = Contrast SS&CP Matrix for Control vs Xpermntl
 E = Error SS&CP Matrix

```
S=1      M=-0.5      N=17
Statistic          Value          F      Num DF   Den DF   Pr > F
Wilks' Lambda      0.867476    5.4997    1        36    0.0247
Pillai's Trace     0.132524    5.4997    1        36    0.0247
Hotelling-Lawley Trace 0.152777    5.4997    1        36    0.0247
Roy's Greatest Root 0.152777    5.4997    1        36    0.0247
```

---<PAGE>-----

The first MANOVA tests for the independent variable therapy. Notice how the F -value here equals that for the simple ANOVA on the difference score given in the previous output. The second MANOVA is the for contrast effect. Once again, the F -value is identical to that in the ANOVA for the contrast effect. The reason for this is obvious--the MANOVA transformation is creating one and only one new variable, the difference between post-test and pre-test scores. Hence, the MANOVA is operating on a 1 by 1 SSCP matrix, which is the definition of an ANOVA.

Profile Analysis


The full value of transformations comes about when one performs several transformations to illuminate the patterning of responses on the dependent variables as a function of the independent variables. One type of transformation is a **polynomial transformation**. This is useful when the dependent variables represent measurements over time. We will treat this in detail in discussion of repeated measures ANOVA in the next chapter. A second important transformation is often called a **profile transformation** that gives rise to a **profile analysis**.

For a profile analysis of many psychological variables where the measurement metric is arbitrary, it is recommended that the dependent variables all be measured on the same scale of measurement. Usually, the most important requirement is that the dependent variables have the same standard deviations, but making their means be the same can aid in interpretation. If they are not measured on the same scale, then PROC

STANDARD may be used to place them on a common metric. To illustrate a profile, consider the following SAS code:

```
TITLE KURLU Example;
TITLE2 Profile Analysis of Difference Scores;
DATA temp;
  SET here.kurlu;
  si_diff = si_post - si_pre;
  sf_diff = sf_post - sf_pre;
  TITLE KURLU Example;
RUN;
PROC SORT; BY THERAPY;
PROC MEANS;
  BY THERAPY;
  VAR si_diff sf_diff oi_diff;
RUN;
```

The means of the three difference scores that come from this output may then be plotted on a graph such as that given below.



There are two attributes to a profile. The first is overall elevation or **level**. Mathematically, this is equal to either the sum or the average of the dependent variables. For the Kurlu data, the profile level for a therapy would be a measure of overall, global improvement.

The second attribute of a profile is its **shape**. Profile shape equals the “hills and valleys” in a plot of the means. Mathematically, profile shape is equal to a series of difference scores. The first difference score is that between the first and second dependent variable, the second is that between the second and third dependent variable, and so on. For the Kurlu data, the profile shape of a therapy is a measure of differential improvement on one outcome measure versus another outcome measure. A test of profile

shape asks whether the four lines in the figure are really parallel to one another except for sampling error. If this test is rejected, then the lines are not parallel.

A profile analysis involves performing two MANOVAs. The first MANOVA transforms the dependent variables into a new variable, level. The second MANOVA transforms them into the new shape variables. As applied to the Kurlu data, the SAS code would be

```
TITLE KURLU Example;
TITLE2 Profile Analysis of Difference Scores;
DATA temp;
  SET here.kurlu;
  si_diff = si_post - si_pre; /* difference scores */
  sf_diff = sf_post - sf_pre;
  oi_diff = oi_post - oi_pre;
RUN;
PROC GLM DATA=temp ORDER=internal;
  CLASS therapy;
  MODEL si_diff sf_diff oi_diff = therapy;
  CONTRAST 'Control vs Exprmntl' therapy -1 -1 -1 3;
RUN;
TITLE3 Profile Level;
MANOVA H=therapy M=si_diff + sf_diff + oi_diff /
  PRINTE;
RUN;
TITLE3 Profile Shape;
MANOVA H=therapy
  M=si_diff - sf_diff,
  sf_diff - oi_diff
  MNAMES = diff1 diff2 /
  PRINTE SUMMARY;
RUN;
```

Some comment is needed here before examining the output. As in the previous examples, there is a contrast between the mean of the three experimental therapies and that of the Control group. In the profile level, the new dependent variable is the sum of the three dependent variables. This gives identical results to those using the average of the three variables in the following MANOVA statement:

```
MANOVA H=therapy
  M=.333*si_diff + .333*sf_diff + .333*oi_diff;
```

The analysis of profile shape creates two new dependent variables. The MANOVA is then performed on these two new variables. The first new variable is the difference between improvement on the Symptom Index and improvement on the Social Functioning measure. The second new dependent variable is difference between the Social Functioning measure and the Occupational Adjustment measure. In the M= option of the MANOVA statement, a comma (,) is used to separate one new variable from the next. In general, if there are q dependent variables, then there will be $(q - 1)$ new dependent variables for a profile shape.

The MNames = option gives names to the two new dependent variables. By default, SAS would name them mvar1 and mvar2, but in this code, they have been called diff1 and diff2. The SUMMARY option requests that SAS print individual ANOVAs for the new variables diff1 and diff2. (Do not forget the slash (/) before the SUMMARY option.) The output from this SAS program is given below.

```

----<Page>-----
KURLU Example
Profile Analysis of Difference Scores
General Linear Models Procedure
Class Level Information

```

```

Class      Levels      Values
THERAPY      4      Abreaction Behavioral Cognitive Control

```

Number of observations in data set = 40

```

----<Page>-----
KURLU Example
Profile Analysis of Difference Scores

```

General Linear Models Procedure

Dependent Variable: SI_DIFF

Source	DF	Sum of Squares	F Value	Pr > F
Model	3	481.40000000	2.27	0.0970
Error	36	2545.00000000		
Corrected Total	39	3026.40000000		

R-Square	C.V.	SI_DIFF Mean
0.159067	102.5366	8.20000000

Source	DF	Type I SS	F Value	Pr > F
THERAPY	3	481.40000000	2.27	0.0970

Source	DF	Type III SS	F Value	Pr > F
THERAPY	3	481.40000000	2.27	0.0970

Contrast	DF	Contrast SS	F Value	Pr > F
Control vs Exprmntl	1	388.80000000	5.50	0.0247

```

----<Page>-----
KURLU Example
Profile Analysis of Difference Scores
General Linear Models Procedure

```

Dependent Variable: SF_DIFF

Source	DF	Sum of Squares	F Value	Pr > F
Model	3	1874.60000000	7.54	0.0005
Error	36	2983.00000000		
Corrected Total	39	4857.60000000		

R-Square	C.V.	SF_DIFF Mean

0.385911 75.22982 12.1000000

Source	DF	Type I SS	F Value	Pr > F
THERAPY	3	1874.60000000	7.54	0.0005

Source	DF	Type III SS	F Value	Pr > F
THERAPY	3	1874.60000000	7.54	0.0005

Contrast	DF	Contrast SS	F Value	Pr > F
Control vs Exprmntl	1	1642.80000000	19.83	0.0001

----<Page>-----

KURLU Example
 Profile Analysis of Difference Scores
 General Linear Models Procedure

Dependent Variable: OI_DIFF

Source	DF	Sum of Squares	F Value	Pr > F
Model	3	293.60000000	1.84	0.1572
Error	36	1914.00000000		
Corrected Total	39	2207.60000000		

R-Square	C.V.	OI_DIFF Mean
0.132995	123.5856	5.90000000

Source	DF	Type I SS	F Value	Pr > F
THERAPY	3	293.60000000	1.84	0.1572

Source	DF	Type III SS	F Value	Pr > F
THERAPY	3	293.60000000	1.84	0.1572

Contrast	DF	Contrast SS	F Value	Pr > F
Control vs Exprmntl	1	34.13333333	0.64	0.4282

----<Page>-----

KURLU Example
 Profile Analysis of Difference Scores
 Profile Level

General Linear Models Procedure
 Multivariate Analysis of Variance

M Matrix Describing Transformed Variables

	SI_DIFF	SF_DIFF	OI_DIFF
MVAR1	1	1	1

----<Page>-----

KURLU Example
 Profile Analysis of Difference Scores
 Profile Level

General Linear Models Procedure
 Multivariate Analysis of Variance

Characteristic Roots and Vectors of: E Inverse * H, where
 H = Type III SS&CP Matrix for THERAPY E = Error SS&CP Matrix

Variables have been transformed by the M Matrix

Characteristic Root	Percent	Characteristic Vector	V'EV=1
		MVAR1	
0.36402007	100.00	0.00842947	

Manova Test Criteria and Exact F Statistics for the Hypothesis of no Overall THERAPY Effect on the variables defined by the M Matrix Transformation
H = Type III SS&CP Matrix for THERAPY E = Error SS&CP Matrix

S=1 M=0.5 N=17

Statistic	Value	F	Num DF	Den DF	Pr > F
Wilks' Lambda	0.733127	4.3682	3	36	0.0101
Pillai's Trace	0.266873	4.3682	3	36	0.0101
Hotelling-Lawley Trace	0.36402	4.3682	3	36	0.0101
Roy's Greatest Root	0.36402	4.3682	3	36	0.0101

Characteristic Roots and Vectors of: E Inverse * H, where
H = Contrast SS&CP Matrix for Control vs Exprmnt1
E = Error SS&CP Matrix

Variables have been transformed by the M Matrix

Characteristic Root	Percent	Characteristic Vector	V'EV=1
		MVAR1	
0.31038223	100.00	0.00842947	

Manova Test Criteria and Exact F Statistics for the Hypothesis of no Overall Control vs Exprmnt1 Effect on the variables defined by the M Matrix Transformation
H = Contrast SS&CP Matrix for Control vs Exprmnt1
E = Error SS&CP Matrix

S=1 M=-0.5 N=17

Statistic	Value	F	Num DF	Den DF	Pr > F
Wilks' Lambda	0.763136	11.174	1	36	0.0019
Pillai's Trace	0.236864	11.174	1	36	0.0019
Hotelling-Lawley Trace	0.310382	11.174	1	36	0.0019
Roy's Greatest Root	0.310382	11.174	1	36	0.0019

---<Page>-----

KURLU Example
Profile Analysis of Difference Scores
Profile Shape
General Linear Models Procedure
Multivariate Analysis of Variance

M Matrix Describing Transformed Variables

	SI_DIFF	SF_DIFF	OI_DIFF
DIFF1	1	-1	0
DIFF2	0	1	-1

E = Error SS&CP Matrix

	DIFF1	DIFF2
DIFF1	2694.2	-1752.5
DIFF2	-1752.5	3184.6

---<Page>-----

KURLU Example
 Profile Analysis of Difference Scores
 Profile Shape
 General Linear Models Procedure
 Multivariate Analysis of Variance

Partial Correlation Coefficients from the Error SS&CP Matrix
 of the Variables Defined by the Specified Transformation / Prob > |r|

DF = 36

	DIFF1	DIFF2
DIFF1	1.000000	-0.598295
	0.0001	0.0001
DIFF2	-0.598295	1.000000
	0.0001	0.0001

---<Page>-----

KURLU Example
 Profile Analysis of Difference Scores
 Profile Shape
 General Linear Models Procedure
 Multivariate Analysis of Variance

Characteristic Roots and Vectors of: E Inverse * H, where
 H = Type III SS&CP Matrix for THERAPY E = Error SS&CP Matrix

Variables have been transformed by the M Matrix

Characteristic Root	Percent	Characteristic Vector		V'EV=1
		DIFF1	DIFF2	
0.38053896	56.88	0.00345919	-0.01563240	
0.28853867	43.12	0.02379366	0.01564319	

Manova Test Criteria and F Approximations for
 the Hypothesis of no Overall THERAPY Effect
 on the variables defined by the M Matrix Transformation
 H = Type III SS&CP Matrix for THERAPY E = Error SS&CP Matrix

S=2 M=0 N=16.5

Statistic	Value	F	Num DF	Den DF	Pr > F
Wilks' Lambda	0.562152	3.8937	6	70	0.0021
Pillai's Trace	0.499572	3.9954	6	72	0.0017
Hotelling-Lawley Trace	0.669078	3.7914	6	68	0.0026
Roy's Greatest Root	0.380539	4.5665	3	36	0.0082

NOTE: F Statistic for Roy's Greatest Root is an upper bound.
 NOTE: F Statistic for Wilks' Lambda is exact.

Characteristic Roots and Vectors of: E Inverse * H, where
 H = Contrast SS&CP Matrix for Control vs Exprmnt1

E = Error SS&CP Matrix

Variables have been transformed by the M Matrix

Characteristic Root	Percent	Characteristic Vector	V'EV=1
		DIFF1	DIFF2
0.37957812	100.00	0.00161758	-0.01679005
0.00000000	0.00	0.02398933	0.01439360

Manova Test Criteria and Exact F Statistics for
the Hypothesis of no Overall Control vs Exprmntl Effect
on the variables defined by the M Matrix Transformation
H = Contrast SS&CP Matrix for Control vs Exprmntl

E = Error SS&CP Matrix

KURLU Example

Profile Analysis of Difference Scores

Profile Shape

General Linear Models Procedure
Multivariate Analysis of Variance

S=1 M=0 N=16.5

Statistic	Value	F	Num DF	Den DF	Pr > F
Wilks' Lambda	0.724859	6.6426	2	35	0.0036
Pillai's Trace	0.275141	6.6426	2	35	0.0036
Hotelling-Lawley Trace	0.379578	6.6426	2	35	0.0036
Roy's Greatest Root	0.379578	6.6426	2	35	0.0036

---<Page>-----

KURLU Example

Profile Analysis of Difference Scores

Profile Shape

General Linear Models Procedure
Multivariate Analysis of Variance

Dependent Variable: DIFF1

Source	DF	Type III SS	F Value	Pr > F
THERAPY	3	901.4000000	4.01	0.0146
Error	36	2694.2000000		

Contrast	DF	Contrast SS	F Value	Pr > F
Control vs Exprmntl	1	433.2000000	5.79	0.0214

Dependent Variable: DIFF2

Source	DF	Type III SS	F Value	Pr > F
THERAPY	3	1205.8000000	4.54	0.0084
Error	36	3184.6000000		

Contrast	DF	Contrast SS	F Value	Pr > F
Control vs Exprmntl	1	1203.3333333	13.60	0.0007

---<Page>-----

In this example, a new data set was created that contained three new variables which were the differences between posttest and pretest for the three measures One

could look at these three new variables as measures of improvement. The first set of MANOVAs is performed on the new dependent variable, level, which is simply the sum of the three improvement variables. The first MANOVA (for H=therapy) tells us whether the means for overall level four groups can be regarded as being sampled from the same “hat” of means vectors. The *F*-ratio is significant, so it is clear that there are some differences among the means.

The next MANOVA is for the contrast. This answers the question of whether the mean level of improvement averaged across the experimental therapies and mean level for the Control can both be sampled from the same “hat.” The *F*-ratio here is highly significant, and from examining the means, there would be good justification to conclude that the experimental therapies, on average, create more improvement than the Control therapy.

The next set of MANOVA are performed on profile shape. Here, there are two new dependent variables. The first of these is the difference in improvement for the Symptom Index and the Social Functioning measure. The second is the difference in improvement between the Social Functioning measure that the Occupational Adjustment Scale. The first MANOVA in this set tests for hypothesis that the profile shape is the same over the four therapies. In other words, are the four lines in the figure parallel? The *F*-ratio is highly significant, so this hypothesis must be rejected. There are indeed, some line(s) that are not parallel to some other line(s).

The final MANOVA tests for similarity in profile shape between the average profile of the three experimental therapies and the profile of the Control. Once again, the *F* is significant, so one concludes that the lines are not parallel.

The last section of the output comes from the SUMMARY option on the MANOVA statement. This will perform two univariate ANOVAs one for each of the transformed variables. The first is for variable Diff1 (difference between Symptom Index improvement and Social Functioning improvement). This is significant for both therapy and for the contrast. The second is for variable Diff2 (difference between Social Functioning improvement and Occupational Adjustment improvement). This is likewise significant for both therapy and the contrast.

A useful exercise would be to run the following program and compare its output to that given above.

```
TITLE KURLU Example;
TITLE2 Profile Analysis of Difference Scores;
DATA temp;
  SET here.kurlu;
  si_diff = si_post - si_pre; /* difference scores */
  sf_diff = sf_post - sf_pre;
  oi_diff = oi_post - oi_pre;
  level = si_diff + sf_diff + oi_diff;
  diff1 = si_diff - sf_diff;
  diff2 = sf_diff - oi_diff;
  IF therapy='Control' THEN convsxp=3;
  ELSE convsxp=-1;
  LABEL convsxp = 'Control vs Expermnt1';
RUN;
```



```

PROC GLM DATA=temp ORDER=internal;
  CLASS therapy;
  MODEL level diff1 diff2 = therapy;
  MANOVA H=therapy
         M=diff1, diff2;
RUN;
PROC GLM DATA=temp;
  MODEL level diff1 diff2 = convsxpr;
  MANOVA H=convsxpr
         M=diff1, diff2;
RUN;

```

A second very good exercise is to go through the following program and then ask what each of these MANOVA statements are equal to in the output that has already been presented. You may also want to run the program to check your answers.

```

LIBNAME here '~carey/p7291dir';
OPTIONS LINESIZE=64 NODATE NOCENTER NONUMBER;
PROC GLM DATA=here.kurlu ORDER=internal;
  CLASS therapy;
  MODEL si_pre sf_pre oi_pre si_post sf_post oi_post
        = therapy / NOUNI;
  CONTRAST 'Control vs Expermnt1' therapy -1 -1 -1 3;
  MANOVA H=therapy
        M=si_pre + sf_pre + oi_pre - si_post - sf_post - oi_post;
  MANOVA H=therapy
        M=si_pre - si_post,
          sf_pre - sf_post,
          oi_pre - oi_post
        MNames=si_diff sf_diff oi_diff /
        SUMMARY;
  MANOVA H=therapy
        M=si_pre - sf_pre - si_post + sf_post,
          sf_pre - oi_pre - sf_post + oi_post
        MNames=prodiffl prodiff2 /
        SUMMARY;
  MANOVA h=therapy
        M=si_pre - oi_pre - si_post + oi_post,
          sf_pre - si_pre - sf_post + si_post
        MNames=prodifx1 prodifx2 /
        SUMMARY;
RUN;

```